Newsgroup: it.comp.lang.visual-basic
Titolo: [?] - Ridimensionamento oggetti

Salve a tutti !
Recentemente ho letto su questo NG una routine per ridimensionare i
controlli contenuti in una form ( textbox, labels, ecc), al cambiare della
risoluzione.
Purtroppo ho perso il msg postato.
Se qualche anima pia lo puņ riproporre ne sarei grato.
Saluti.
--
Genesio Conti - Calvizzano (NA) - Italy
E-Mail :  gen...@NOSPAMsofthome.net
( remove NOSPAM for answer )

C'č un modo per ridimensionarli proporzionalmente alla larghezza del form?
Prova a mettere questo codice nell'evento resize del form:
treview.move 0,0,me.scalewidth* N  ,me.scaleheight
Rich.move treview.width,0,me.scalewidth-treview.width,me.scaleheight
N sta per l'aria che dovra ricoprire la treview nel form.
Es:
N=0.5 'metą form
n=0.3 'un terzo del form
n=0.6 'due terzi del form
n=1 'tutto il form
Spero di esserti stato di aiuto
Ciao
Amedeo

 Amedeo
Madonna, datemi il tempo di sciverlo il msg... hi hi
Ué grazie...
Genesio

Private Declare Function EnumDisplaySettings Lib "user32" Alias "EnumDisplaySettingsA" (ByVal lpszDeviceName As Long,
ByVal iModeNum As Long, lpDevMode As Any) As Boolean
Private Declare Function ChangeDisplaySettings Lib "user32" Alias "ChangeDisplaySettingsA" (lpDevMode As Any, ByVal
dwflags As Long) As Long
       Const CCDEVICENAME = 32
       Const CCFORMNAME = 32
       Const DM_PELSWIDTH = &H80000
       Const DM_PELSHEIGHT = &H100000
Private Type DEVMODE
       dmDeviceName As String * CCDEVICENAME
       dmSpecVersion As Integer
       dmDriverVersion As Integer
       dmSize As Integer
       dmDriverExtra As Integer
       dmFields As Long
       dmOrientation As Integer
       dmPaperSize As Integer
       dmPaperLength As Integer
       dmPaperWidth As Integer
       dmScale As Integer
       dmCopies As Integer
       dmDefaultSource As Integer
       dmPrintQuality As Integer
       dmColor As Integer
       dmDuplex As Integer
       dmYResolution As Integer
       dmTTOption As Integer
       dmCollate As Integer
       dmFormName As String * CCFORMNAME
       dmUnusedPadding As Integer
       dmBitsPerPel As Integer
       dmPelsWidth As Long
       dmPelsHeight As Long
       dmDisplayFlags As Long
       dmDisplayFrequency As Long
End Type
Dim DevM As DEVMODE
Sub ChangeRes(iWidth As Single, iHeight As Single)
       Dim a As Boolean
       Dim i&
       i = 0
              Do
                     a = EnumDisplaySettings(0&, i&, DevM)
                     i = i + 1
              Loop Until (a = False)
       Dim b&
       DevM.dmFields = DM_PELSWIDTH Or DM_PELSHEIGHT
       DevM.dmPelsWidth = iWidth
       DevM.dmPelsHeight = iHeight
       b = ChangeDisplaySettings(DevM, 1)
End Sub
Sub SetReso(frm As Form, RO!, RV!)
   On Error GoTo ErrH
' dimensione della form
    frm.Width = frm.Width * RO
    frm.Height = frm.Height * RV
' dimensione e posizione dei control
For i = 0 To frm.Controls.Count - 1
    frm.Controls(i).Left = frm.Controls(i).Left * RO
     frm.Controls(i).Top = frm.Controls(i).Top * RV
    frm.Controls(i).Width = frm.Controls(i).Width * RO
    frm.Controls(i).Height = frm.Controls(i).Height * RV
    frm.Controls(i).FontSize = frm.Controls(i).FontSize * RV
Next i
'posizione eventuali linee
   For i = 0 To frm.Linea.Count - 1
      frm.Linea(i).X1 = frm.Linea(i).X1 * RO
      frm.Linea(i).X2 = frm.Linea(i).X2 * RO
      frm.Linea(i).Y1 = frm.Linea(i).Y1 * RV
      frm.Linea(i).Y2 = frm.Linea(i).Y2 * RV
   Next
Exit Sub
ErrH:
    Resume Next
End Sub
Utilizzo:
Memorizzo in due variabili i valori di risoluzione attuali:
AttualeX=Screen.Width / Screen.TwipsPerPixelX
AttualeY=Screen.Height / Screen.TwipsPerPixelY
Poi cambio risoluzione:
ChangeRes(1024, 768)
E ridimensiono i controlli:
RapportoOres = (Screen.Width / Screen.TwipsPerPixelX) / AttualeX
RapportoVres = (Screen.Height / Screen.TwipsPerPixelY) / AttualeY
Call SetReso(me, RapportoOres, RapportoVres)
Ciao,
Moreno
#sirri#@
tin.it
(togliete "#" dall'indirizzo)
---------------------------------------------------------------------------
Le risposte alle tue domande le trovi sul Sito Comune di questo NewsGroup:
 
http://www.murialdo.it/it_lang_vb
---------------------------------------------------------------------------

Grazie Moreno, era proprio questo msg che ricordavo...
Grazie comunque anche ad Amedeo, anche le tue info mi sono utili...
